Optimization

Lesson 2.2

Formulation Exercises

Formulating Linear Problems

A product-mix problem

- A furniture manufacturer produces tables and chairs. The process involves machining, sanding, and assembling the pieces to make the tables and chairs.
- It takes **5h** to machine the pieces for a table, **4h** to sand the pieces, and **3h** to assemble a table.
- A chair requires **2h** to machine the pieces, **3h** to sand the pieces, and **4h** to assemble a chair.
- There are **270h** available for machining the pieces, **250h** for sanding the pieces, and **200h** for assembling.
- If the profit for a table is **€100** and for a chair **€60**, how many tables and chairs should the manufacturer produce in order to maximize the overall profit?
- What if there is an additional requirement to produce four chairs for each table?

LP Model

Resource/Item	Per Unit Product		Resource Availability (h)
	Table	Chair	
Machining	5	2	270
Sanding	4	3	250
Assembly	3	4	200
Profit per unit (€)	100	60	

$$\text{Maximize } Z = 100T + 60C$$

Subject to

$$5T + 2C \leq 270 \quad \text{Machining}$$
$$4T + 3C \leq 250 \quad \text{Sanding}$$
$$3T + 4C \leq 200 \quad \text{Assembly}$$
$$T \geq 0 \text{ and } C \geq 0 \quad \text{Nonnegativity}$$

Excel Model

Excel Model

Decision variables	Table	Chair	Objective function
Number of units to produce	0	0	0
Profit per unit	100	60	

Constraints			LHS	RHS
Machining	5	2	0	270
Sanding	4	3	0	250
Assembly	3	4	0	200



Excel Solution

Decision variables	Table	Chair		Objective function
Number of units to produce	48.5714	13.5714		5671.428571
Profit per unit	100	60		

Constraints			LHS	RHS
Machining	5	2	270	270
Sanding	4	3	235	250
Assembly	3	4	200	200

Objective Cell (Max)

Cell	Name	Original Value	Final Value
$L$11	Number of units to produce Objective function	0	5671.428571

Variable Cells

Cell	Name	Original Value	Final Value	Integer
$I$11	Number of units to produce Table	0	48.57142857	Contin
$J$11	Number of units to produce Chair	0	13.57142857	Contin

Constraints

Cell	Name	Cell Value	Formula	Status	Slack
$K$15	Machining LHS	270	$K$15<=$L$15	Binding	0
$K$16	Sanding LHS	235	$K$16<=$L$16	Not Binding	15
$K$17	Assembly LHS	200	$K$17<=$L$17	Binding	0

IBM ILOG CPLEX Optimization Studio

How to install IBM CPLEX

- Go to https://www.ibm.com/products/ilog-cplex-optimization-studio
- Choose “No-cost academic edition”
- Sign up with the FEUP email address
- Fill the form and check the email to complete the registration



How to install IBM CPLEX

- Choose the packages you want to download

1. Creating a new project



2. Adding a run configuration



3. Creating a model





3.1 Model 1

File .mod – with the model

TablesChairs_Model1.mod

```
/*********************************************
 * OPL 20.1.0.0 Model
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 14:06:35
 *********************************************/
//declare data

{string} Products =...; // products

float profit[Products] = ...;
float machining[Products] = ...;
float sanding[Products] = ...;
float assembly[Products] = ...;
float timeMachining = ...;
float timeSanding = ...;
float timeAssembly = ...;

//decision variables
dvar float+ x[Products];

maximize sum(i in Products) profit[i] * x[i] ;

subject to {
   sum (i in Products) machining[i]* x[i] <= timeMachining;
   sum (i in Products) sanding[i]* x[i] <= timeSanding;
   sum (i in Products) assembly[i]* x[i] <= timeAssembly;
 }

 execute OUTPUT_RESULTS {
   var file = new IloOplOutputFile ("solutionTablesChairs1.txt", true);
   file.writeln ("Objective Function = ", cplex.getObjValue ());
   for (i in Products)
    file.writeln ("Variable x[", i, "] = ", x[i]);
  }
```

File .dat – with the data

TablesChairs_Model1.mod | TablesChairs_Data1.dat

```
/*********************************************
 * OPL 20.1.0.0 Data
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 15:38:49
 *********************************************/

Products = {"table","chair"};
profit= [100,  60];
machining = [5 2] ;
sanding = [4,3] ;
assembly = [3,4] ;
timeMachining = 270;
timeSanding = 250;
timeAssembly = 200;
```

3.1 Running Model 1





3.2 Model 2

File .mod - with the model

TablesChairs_Model2.mod | TableChairs_Data2.dat

```
/*********************************************
 * OPL 20.1.0.0 Model
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 17:52:22
 *********************************************/
//declare data

{string} Products =...; // products
{string} Sections =...; // sections

float Profit[Products] = ...;
float TimePerSection[Sections][Products] = ...;

float AvailableTime[Sections] = ...;

//decision variables
dvar float+ x[Products];

maximize sum(i in Products) Profit[i] * x[i] ;

subject to {
  forall (i in Sections)
    sum (j in Products) TimePerSection[i,j]* x[j] <= AvailableTime[i];
 }

 execute OUTPUT_RESULTS {
   var file = new IloOplOutputFile ("solutionTablesChairs1.txt", true);
   file.writeln ("Objective Function = ", cplex.getObjValue ());
   for (i in Products)
    file.writeln ("Variable x[", i, "] = ", x[i]);
  }
```

The sections are not defined in the model

File .dat - with the data

TablesChairs_Model2.mod | TableChairs_Data2.dat

```
/*********************************************
 * OPL 20.1.0.0 Data
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 18:01:58
 *********************************************/
Products = {"table","chair"};
Sections = {"machining","sanding","assembly"};

Profit= [100,  60];
TimePerSection = [[5,2],[4,3],[3,4]];
AvailableTime =[270,250,200];
```

To run this new model we will need to create a new Run Configuration



3.3 Model 3 – reading data from a excel file

File .mod is the same of Model 1 (for simplicity)

TablesChairs_Model1.mod | TablesChairs_DataExcel1.dat

```
/*********************************************
 * OPL 20.1.0.0 Model
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 14:06:35
 *********************************************/
//declare data

{string} Products =...; // products

float profit[Products] = ...;
float machining[Products] = ...;
float sanding[Products] = ...;
float assembly[Products] = ...;
float timeMachining = ...;
float timeSanding = ...;
float timeAssembly = ...;

//decision variables
dvar float+ x[Products];

maximize sum(i in Products) profit[i] * x[i] ;

subject to {
   sum (i in Products) machining[i]* x[i] <= timeMachining;
   sum (i in Products) sanding[i]* x[i] <= timeSanding;
   sum (i in Products) assembly[i]* x[i] <= timeAssembly;
 }
```

File .dat – links to the excel file

TablesChairs_Model1.mod | TablesChairs_DataExcel1.dat

```
/*********************************************
 * OPL 20.1.0.0 Data
 * Author: tgalv
 * Creation Date: 10 Feb 2022 at 16:50:16
 *********************************************/
//n=4;
//m=3;

SheetConnection my_sheet("TablesChairs_dataExcel1.xlsx");

Products from SheetRead(my_sheet, "Products");
machining from SheetRead(my_sheet, "Machining");
sanding from SheetRead(my_sheet, "Sanding");
assembly from SheetRead(my_sheet, "Assembly");
profit from SheetRead(my_sheet, "Profit");

timeMachining from SheetRead(my_sheet, "TimeMachining");
timeSanding from SheetRead(my_sheet, "TimeSanding");
timeAssembly from SheetRead(my_sheet, "TimeAssembly");

x to SheetWrite(my_sheet, "results");
```

3.3 Model 3 – reading data from a excel file



In the excel file we need to define names for the cells that characterize the problem

	A	B	C	D	E	F	G
1	Products	chair	table		AvailableTime		
2	Machining	5	2		270		
3	Sanding	4	3		250		
4	Assembly	3	4		200		
5	Profit	100	60				
6							
7	results	48.57143	13.57143				
8							
9							
10							



3.4 Model 4 – Exercise

Add the new constraint (4 chairs for each table) and solve the integer problem

(i) in excel

(ii) in CPLEX

Homework Assignment

Problem 1 - Corporation Weigelt

- The corporation Weigelt has three factories with excess of production capacity. Fortunately, the company has a new product ready to be produced, and all three factories can produce it. Thus, some of the excess of capacity can be used for the new product.
- This product can be produced in three different sizes (large, medium and small), with a unitary profit of 420€, 360€ and 300€, respectively. Factories 1, 2 and 3 have excess of capacity, which enables them to produce 750, 900 and 450 units of this product per day, respectively, independently of the produced size or the combination of different sizes.
- The availability of storage space during the production process also imposes a limitation in the production of the new product. Factories 1, 2 and 3 have storage capacity of 13 000, 12 000 and 5 000 square meters, respectively, for a production day. Each unit of large, medium and small product produced per day requires 20, 15 and 12 square meters, respectively.
- The sales forecast indicates that 900, 1200 and 750 units of large, medium and small sizes can be sold daily, respectively.
- The management wants to know how many units of each size to produce in each factory in order to maximize the profit.

Problem 1 - Formulation

Decision Variables: x_{ij} = number of products produced in factory i with size j

Objective function:

$$\text{Max } Z = 420\,x_{11} + 360\,x_{12} + 300\,x_{13} + 420\,x_{21} + 360\,x_{22} + 300\,x_{23} + 420\,x_{31} + 360\,x_{32} + 300\,x_{33}$$

Constraints:

$$x_{11} + x_{12} + x_{13} \leq 750$$
$$x_{21} + x_{22} + x_{23} \leq 900$$
$$x_{31} + x_{32} + x_{33} \leq 450$$

→ Production capacity

$$20\,x_{11} + 15\,x_{12} + 12\,x_{13} \leq 13000$$
$$20\,x_{21} + 15\,x_{22} + 12\,x_{23} \leq 12000$$
$$20\,x_{31} + 15\,x_{32} + 12\,x_{33} \leq 5000$$

→ Storage capacity

$$x_{11} + x_{21} + x_{31} \leq 900$$
$$x_{12} + x_{22} + x_{32} \leq 1200$$
$$x_{13} + x_{23} + x_{33} \leq 750$$

→ Sales forecast

$x_{ij} >= 0$, i,j =1,..3

Problem 1 – Excel

xij	number of units produced in factory i with size j									
		Product sizes								
		Size L	Size M	Size S	Production /Factory		Production Capacity	Storage occupied		Storage Space
Factories	Factory 1	0	0	0	0	<=	750	0	<=	13000
	Factory 2	0	0	0	0	<=	900	0	<=	12000
	Factory 3	0	0	0	0	<=	450	0	<=	500
	Space required	20	15	12						
	Sales	0	0	0						
		<=	<=	<=						
	Sales forecast	900	1200	750			Total profit			
	Unitary Profit	€ 420.00	€ 360.00	€ 300.00			€ -			

Problem 1 - CPLEX

File weigelt.mod

```
{string} Factories =...; // factories
{string} Prod_Sizes =...; // product sizes


float Profit[Prod_Sizes] = ...;
float SalesForecast [Prod_Sizes] = ...;
float SpaceRequired[Prod_Sizes] = ...;
float SpaceAvailable[Factories]= ...;
float ProdCapacity[Factories]= ...;

//decision variables
dvar float+ x[Factories][Prod_Sizes];

maximize sum(j in Prod_Sizes) sum(i in Factories) Profit[j] * x[i][j] ;
 subject to {
   // production capacity
  forall (i in Factories)
   sum (j in Prod_Sizes) x[i][j] <= ProdCapacity[i];
   //storage capacity
  forall (i in Factories)
   sum (j in Prod_Sizes) SpaceRequired[j]*x[i][j] <= SpaceAvailable[i];
   //sales
  forall (j in Prod_Sizes)
   sum (i in Factories) x[i][j] <= SalesForecast[j];

 }
```

File weigelt.dat

```
SheetConnection my_sheet("Weigelt1.xlsx");

Factories from SheetRead(my_sheet, "Factories");
Prod_Sizes from SheetRead(my_sheet, "Prod_Sizes");
ProdCapacity from SheetRead(my_sheet, "ProdCapacity");

Profit from SheetRead(my_sheet, "Profit");
SalesForecast from SheetRead(my_sheet, "SalesForecast");
SpaceAvailable from SheetRead(my_sheet, "SpaceAvailable");
SpaceRequired from SheetRead(my_sheet, "SpaceRequired");


x to SheetWrite(my_sheet, "results");
```

Use the previous excel file, but do not forget to define names for the parameters of the problem

Problem 1 - Corporation Weigelt

A new constraint:

In each factory, some of the employees may be fired, unless most of the excess production could be used to produce the new product. To avoid firing employees, management decided that the three factories should use the same percentage of the excess capacity to produce the new product.

Problem 1 - Formulation

New constraints:

$$\frac{1}{750}(x_{11} + x_{12} + x_{13}) - \frac{1}{900}(x_{21} + x_{22} + x_{23}) = 0$$
$$\frac{1}{750}(x_{11} + x_{12} + x_{13}) - \frac{1}{450}(x_{31} + x_{32} + x_{33}) = 0$$

We could also include the following constraint, but it is redundant considering the last 3 equations

$$\frac{1}{900}(x_{21} + x_{22} + x_{23}) - \frac{1}{450}(x_{31} + x_{32} + x_{33}) = 0$$

Homework: adapt the excel file (and OPL project) to these new constraints. Comment the results